## EXAMPLE 9.4-2: Evaporation from a Lake

You have been asked to estimate the rate at which water evaporates from the surface of a small lake. The lake is roughly circular with a diameter of $D$ = 1000 m. The temperature of the water surface is about $T_s$ =12°C. Wind is blowing over the lake with temperature $T_\infty$ = 18°C and relative humidity $RH_\infty$ = 45%. The wind velocity is not known; however, a conservative estimate of the wind velocity is $u_\infty$ = 1 m/s, which perhaps is low enough to not significantly disturb the lake surface.

a.) Estimate the rate at which water evaporates from the lake surface.

The input information is entered in EES:

```
"EXAMPLE 9.4-2: Evaporation from a Lake"
$UnitSystem SI MASS RAD PA K J
$Tabstops 0.2 0.4 0.6 3.5 in

D = 1000 [m]                              "diameter of the lake"
T_s=converttemp(C,K,12 [C])               "temperature of the lake surface"
T_infinity=converttemp(C,K,18 [C])        "temperature of the wind"
RH_infinity=0.45 [-]                      "relative humidity of the wind"
u_infinity=1 [m/s]                        "estimate of the wind velocity"
p=1 [atm]*convert(atm,Pa)                 "pressure"
```

A heat transfer analogy for flow over an isothermal flat plate will be employed to estimate the evaporation rate from the lake. The Reynolds number and Schmidt number will be determined and used in the Nusselt number correlation for flow over a flat plate, provided in Section 4.9.2, in order to compute the Sherwood number. The film temperature is computed:

$$T_{film} = \frac{T_s + T_\infty}{2}$$

and used to determine the required air properties ($\mu$, $\rho$, and $\nu$) using EES' built-in property routine for air. The diffusion coefficient for water vapor in air ($D_{a,w}$) is estimated using the EES function D_12_gas.

```
T_film=(T_s+T_infinity)/2                 "film temperature"
mu=viscosity(Air,T=T_film)                "viscosity"
rho=density(Air,T=T_film,p=p)             "density"
nu=mu/rho                                 "kinematic viscosity"
D_a_w=D_12_gas('Air','Water',T_film,p)    "diffusion coefficient"
```

The Schmidt number is computed according to:

$$Sc = \frac{\upsilon}{D_{a,w}}$$

and the Reynolds number is computed according to:

$$Re = \frac{\rho D u_\infty}{\mu}$$

where the characteristic length of the lake is assumed to be its diameter. The convection correlation for flow over an isothermal plate is accessed using the EES function External_Flow_Plate_ND; note that input Prandtl number is replaced with the Schmidt number and the output is assigned to the average Sherwood number ( $\overline{Sh}$ ) rather than the average Nusselt number, as indicated by Eq. (9-88).

```
Sc=nu/D_a_w                                    "Schmidt number"
Re=rho*D*u_infinity/mu                         "Reynolds number"
Call External_Flow_Plate_ND(Re,Sc: Sh_bar,C_f)
      "obtain Sherwood number using external convection correlation for a flat plate"
```

The average Sherwood number is used to compute the average mass transfer coefficient:

$$\overline{h}_D = \frac{\overline{Sh}\, D_{a,w}}{D}$$

```
h_D_bar=D_a_w*Sh_bar/D        "mass transfer coefficient"
```

The mass transfer rate is driven by the difference between the concentration of water vapor at the lake surface and in the free stream. The partial pressure of the water vapor at the lake surface is the saturation pressure of water at $T_s$ ($p_{w,sat}$), evaluated using an EES property routine. The concentration of water vapor at the lake surface ($c_{w,sat}$) is the density of water vapor evaluated at the partial pressure and temperature. The mass fraction of water vapor at the lake surface is:

$$mf_{w,sat} = \frac{c_{w,sat}}{\rho}$$

```
p_w_sat=pressure(Water,x=1,T=T_s)         "saturation pressure of water vapor at the lake surface"
c_w_sat=density(Water,p=p_w_sat,x=1)      "concentration of water vapor at lake surface"
mf_w_sat=c_w_sat/rho                      "mass fraction of water vapor at the lake surface"
```

The partial pressure of water in the free stream ($p_{w,\infty}$) is the product of the relative humidity and the saturation pressure of water evaluated at $T_\infty$ ($p_{w,sat,\infty}$) evaluated using the EES property routine.

$$p_{w,\infty} = RH\, p_{sat,w,\infty}$$

The concentration of water vapor in the free stream ($c_{w,\infty}$) is the density of water evaluated at the partial pressure and temperature. The mass fraction of water in the free stream is:

$$mf_{w,\infty} = \frac{c_{w,\infty}}{\rho}$$

```
p_w_sat_infinity=pressure(Water,x=1,T=T_infinity)       "sat. pressure of water vapor in the free stream"
p_w_infinity=RH_infinity*p_w_sat_infinity               "partial pressure of water vapor in the free stream"
c_w_infinity=density(Water,p=p_w_infinity,T=T_infinity) "concentration of water vapor in the free stream"
```

The blowing factor is calculated using Eqs. (9-89) and (9-90):

$$BF = \frac{\ln(1+B)}{B}$$

$$B = \frac{mf_{w,\infty} - mf_{w,sat}}{mf_{w,sat} - 1}$$

```
B=(mf_w_infinity-mf_w_sat)/(mf_w_sat-1)      "B for blowing factor"
BF=ln(1+B)/B                                 "blowing factor"
```

which leads to $BF$ = 0.9985; the mass fraction of water in air is small and therefore the correction associated with the induced velocity at the surface of the lake is negligible. The corrected mass transfer coefficient is:

$$\overline{h}_{D,c} = \overline{h}_D \, BF$$

```
h_D_c_bar=h_D_bar*BF          "mass transfer coefficient, corrected for blowing"
```

The mass flow rate of water due to evaporation is calculated according to Eq. (9-83) using the corrected mass transfer coefficient:

$$\dot{m}_w = \overline{h}_{D,c} \frac{\pi D^2}{4} \left( c_{w,sat} - c_{w,\infty} \right)$$

The volume rate at which liquid water evaporates from the lake is given by:

$$\dot{V}_w = \frac{\dot{m}_w}{\rho_{w,l}}$$

where $\rho_{w,l}$ is the density of liquid water, evaluated using the EES property function.

```
m_dot_w=h_D_bar*(pi*D^2/4)*(c_w_sat-c_w_infinity)     "evaporation mass flow rate"
rho_w_l=density(Water,T=T_s,x=0)                      "density of liquid water"
V_dot_w=m_dot_w/rho_w_l                               "evap. volumetric flow rate of liq. water"
V_dot_w_gpd=V_dot_w*convert(m^3/s,gal/day)            "in gallon per day"
```

which leads to $\dot{m}_w$ = 3.58 kg/s and $\dot{V}_w$ = 3.6x10$^{-3}$ m$^3$/s (82,000 gal/day). This is obviously a very approximate calculation, but how else could you obtain this estimate? One value of this method is that it correctly predicts trends. For example, Figure 1 shows the predicted rate of liquid loss as a function of the relative humidity in the air. With the assumed lake and air temperatures, evaporation loss will stop when the relative humidity is 0.7 and condensation on the lake surface will begin to occur at higher relative humidities.




**Figure 1: Estimated rate of liquid loss as a function of relative humidity**